
*                                                                        
*                                           ^  VC
*               _____              R3      /_\       R4
*              |     |         _/\  /\  /\__|_/\  /\  /\_          _____
*              |    _|_       |   \/  \/        \/  \/   |        |     |
*             _|_  / _ \IB1   |                          |       _|_    |
*             \\\  \/ \/      |________          ________|   IB2/ _ \  _|_
*                  /\_/\              |         |               \/ \/  \\\
*                  \___/              |VCN3     |               /\_/\
*                    |        VBN1    |         |       VBN2    \___/
*              ^ VC  |___________     |     VCN4|      ___________|
*             /_\    | QN1 |  QN2|    |         |     |QN5   |QN6 |
*             _|_    |    _|    _|   _|         |_    |_     |_   |
*            /VCC\   |__|'____|'___|'   QN3  QN4  `|____`|_____`|_|
*            \___/      |`->  |`-> |`->         <-'|  <-'|   <-'|
*              |        MX |  1X |    |NX     NX|     |1X    |MX
*             _|_          |     |    |         |_____|      |
*             ///          |     |____|_____     VEN4 |      |
*                          |      VEN3      |         |      |
*                          |_______________/|\________|      |
*                          |                |________________|
*                VBP!     <-                                   ->
*                  ____|' QP1           OUT                 QP2 `|_
*       VIN      _|_   |`_               _/\  /\  /\_         _ '|_|_
*        ______ /   \     _|_     VCN1 _|_  \/  \/  _|_      |    ///
*       |VAC1  |EGin |    ///     ___ /   \   ROUT  ///     _|_
*      _|_    _|     |            ___| GDIF|                \\\
*     /AC1\   | \___/            VCN2 \___/
*     \___/   |___|                    _|_
*       |        _|_                   \\\
*      _|_       ///
*      \\\
*
*

.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VC     0             DC     10
VAC1      VIN    0             DC     0      SIN(   0      1   1000 )
IB1       0      VBN1          1u
IB2       0      VBN2          1u
QN1       VBN1   VBN1   VE4    NPN1   4.70
QN2       VBN1   VBN1   VE3    NPN1   1.00
QN3       VCN3   VBN1   VE3    NPN1   15.00

QN4       VCN4   VBN2   VE4    NPN1   15.00
QN5       VBN2   VBN2   VE4    NPN1   1.00
QN6       VBN2   VBN2   VE3    NPN1   4.70

QP1       0      VBP1   VE3    PNP1   1.00
QP2       0      0      VE4    PNP1   1.00
R3        VCN3   VC     1K
R4        VCN4   VC     1K
E_DIF     OUT    0      VCN3   VCN4   1
ROUT      OUT    O      1K
E_GAININ  VBP1   0      VIN    0      1m

.control
tran             25u                   5m         0    1u
plot             out
echo             "THD% versus VIN_vpk "
setplot          new
set              NameList =  (        Mx4 Mx4pt25 Mx4pt5 Mx5  )
compose          NxVals      values   4   4.25    4.5    5
compose          VinVals     values   10m 30m 35m 40m 60m 70m 80m 100m 110m 120m 140m 150m 170m 185m 200m  
250m
settype          voltage     VinVals
let              NoOfNx  =   length(NxVals)
let              NoOfVin =   length(VinVals)

begin
unset            interrupt
* ===============Loop_Nx============
let              j         =  1
while            (j       <=  NoOfNx )
let              Nx        =  NxVals[j-1]
alter            QN1  area =  $&Nx 
alter            QN6  area =  $&Nx 
set              thisName  =  $NameList[$&j]
let              $thisName =  0*vector(NoOfVin)



* ===============Loop_Vin==============
let              k             = 1
while            (k           <= NoOfVin )
let              Vin           = VinVals[k-1]
alter            e_gainin gain = $&Vin
tran             25u              5m        0    1u
linearize
set              specwindow=   "blackman"
spec             200    8k      200     v(out)
let              thdsq =mag(out[9])^2 +mag(out[14])^2  +mag(out[19])^2  +mag(out[24])^2
let              thd_percent= 100*sqrt(thdsq)/mag(out[4]) 
echo             "$&unknown.Vin       $&thd_percent"
let              unknown.{$thisName}[unknown.k-1] = thd_percent
repeat           3
destroy
end
if               ($?interrupt)
goto             bail
endif
let              k =          k + 1
endwhile
setscale         VinVals
plot             $NameList loglog title "AB_BIAS THD_% vs Vin_pK and Mx"
let              j =          j + 1
endwhile
label            bail
echo            "Done."
end
.endc

*======== ====== ====== ====== ====== ====== ====== ====== ======
.model    NPN1   NPN(   BF=2100 VAF=216 )
.model    PNP1   PNP(   BF=2100  VAF=21)
.end

=====================END_OF_SPICE=======================

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm

This simulation only works on MacSpice for now.
Data in spice apparently gets stored in vectors which
are ready to be plotted. 

The real goal to a OTA differential input is to actually be 
able to lower the overall offset and noise while at the
same time improving the distortion performance. 

A normal input stage actually has its noise defined by the 
number of electrons that flow through it. To be able to break
this barrier, the input stage needs to be AB biased. That means
that the differential input stage needs to be able to put out
more differential output current than it is drawing DC wise.  

A invention above is doing this with two area ratios. The NX ratio
term defines a current gain which is possible to take place in
output transistors QN3 and QN4. The MX term defines what DC
current is present with no input signal. 

This simulation tests various sizes of the M area ratio.
The larger the value of M, the more the AB_Biased input 
stage is B biased. Less current will flow with no signal.
This will lower the overall effects of both offset and 
noise. However the input distortion will also be higher 
because it will introduce a type of crossover distortion.

A smaller value of M is a more A biased input stage. The
distortion at low signals is better. But the performance
of noise, offset, and maximum input signal are all reduced. 

For a M value around 4.5, the 1% distort is about at the
150mV level compared to the 18mV level for a normal 
differential input stage. The distortion levels for the 
dual differential input stage are shown below.



There appears to be least a factor of two increase in 
the magnitude of input voltage signal. But really the dual
differential input stage should be thought of as having 
1% distortion at 62% of its maximum differential output 
current. The AB_Bias input is at about 75% of the 
maximum available output current at the same level
of distortion. 


